<commit_before>{ stdenv, fetchFromGitHub, fetchurl, fetchpatch
, curl, ed, pkgconfig, zlib, libX11
, SDL2, SDL2_image, SDL2_mixer
}:

stdenv.mkDerivation rec {
  pname = "redeclipse";
  version = "1.6.0";

  src = fetchurl {
    url = "https://github.com/red-eclipse/base/releases/download/v${version}/redeclipse_${version}_nix.tar.bz2";
    sha256 = "0j98zk7nivdsap4y50dlqnql17hdila1ikvps6vicwaqb3l4gaa8";
  };

  buildInputs = [
    libX11 zlib
    SDL2 SDL2_image SDL2_mixer
  ];

  nativeBuildInputs = [
    curl ed pkgconfig
  ];

  makeFlags = [ "-C" "src/" "prefix=$(out)" ];

  patches = [
    # "remove gamma name hack" - Can't find `____gammaf128_r_finite` otherwise
    # Is likely to be included in next release
    (fetchpatch {
       url = "https://github.com/red-eclipse/base/commit/b16b4963c1ad81bb9ef784bc4913a4c8ab5f1bb4.diff";
       sha256 = "1bm07qrq60bbmbf5k9255qq115mcyfphfy2f7xl1yx40mb9ns65p";
    })
  ];

  enableParallelBuilding = true;

  installTargets = [ "system-install" ];

  postInstall = ''
      cp -R -t $out/share/redeclipse/data/ data/*
  '';

  meta = {
    description = "A first person arena shooter, featuring parkour, impulse boosts, and more.";
    longDescription = ''
      Red Eclipse is a fun-filled new take on the first person arena shooter,
      featuring parkour, impulse boosts, and more. The development is geared
      toward balanced gameplay, with a general theme of agility in a variety of
      environments.
    '';
    homepage = https://www.redeclipse.net;
    license = with stdenv.lib.licenses; [ zlib cc-by-sa-30 ];
    maintainers = with stdenv.lib.maintainers; [ lambda-11235 ];
    platforms = stdenv.lib.platforms.linux;
    hydraPlatforms = [];
  };
}<commit_msg>redeclipse: fix license info<commit_after>{ stdenv, fetchFromGitHub, fetchurl, fetchpatch
, curl, ed, pkgconfig, zlib, libX11
, SDL2, SDL2_image, SDL2_mixer
}:

stdenv.mkDerivation rec {
  pname = "redeclipse";
  version = "1.6.0";

  src = fetchurl {
    url = "https://github.com/red-eclipse/base/releases/download/v${version}/redeclipse_${version}_nix.tar.bz2";
    sha256 = "0j98zk7nivdsap4y50dlqnql17hdila1ikvps6vicwaqb3l4gaa8";
  };

  buildInputs = [
    libX11 zlib
    SDL2 SDL2_image SDL2_mixer
  ];

  nativeBuildInputs = [
    curl ed pkgconfig
  ];

  makeFlags = [ "-C" "src/" "prefix=$(out)" ];

  patches = [
    # "remove gamma name hack" - Can't find `____gammaf128_r_finite` otherwise
    # Is likely to be included in next release
    (fetchpatch {
       url = "https://github.com/red-eclipse/base/commit/b16b4963c1ad81bb9ef784bc4913a4c8ab5f1bb4.diff";
       sha256 = "1bm07qrq60bbmbf5k9255qq115mcyfphfy2f7xl1yx40mb9ns65p";
    })
  ];

  enableParallelBuilding = true;

  installTargets = [ "system-install" ];

  postInstall = ''
      cp -R -t $out/share/redeclipse/data/ data/*
  '';

  meta = with stdenv.lib; {
    description = "A first person arena shooter, featuring parkour, impulse boosts, and more.";
    longDescription = ''
      Red Eclipse is a fun-filled new take on the first person arena shooter,
      featuring parkour, impulse boosts, and more. The development is geared
      toward balanced gameplay, with a general theme of agility in a variety of
      environments.
    '';
    homepage = "https://www.redeclipse.net";
    license = with licenses; [ licenses.zlib cc-by-sa-30 ];
    maintainers = with maintainers; [ lambda-11235 ];
    platforms = platforms.linux;
    hydraPlatforms = [];
  };
}
{'fraction_non_alphanumeric': 0.11602527283170591, 'fraction_numerical': 0.049971280873061456, 'mean_word_length': 3.3428927680798006, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 6, 'lorem ipsum': 0, 'www.': 2, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 2, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '2211005', 'n_tokens_mistral': 1343, 'n_tokens_neox': 1223, 'n_words': 277}